
Form f_MatchNamesGifts module
Option Compare Database

Option Explicit

Private Sub cmdMatchNamesGifts_Click()

'This routine goes through and selects all the Names that have been selected.

'A separate subroutine (UpdateNameGiftTable) will actually put the data in the correct table.

Dim i As Variant  'Doesn't like this being an integer and I don't know why!

'loops through all of the selected items, calling the NameID i

'(for example if Abby is selected, her AdopteeID is 6, so i would equal 6 the first time through

' if Christine is also selected, the next time through i would be 8)

For Each i In lboSelectName.ItemsSelected

    UpdateNameGiftTable lboSelectName.ItemData(i)

Next

MsgBox ("All information is entered!")

End Sub

Private Sub UpdateNameGiftTable(AdopteeID)

'This depends on having Microsoft DAO 3.6 Object library and Microsoft ActiveX Data Objects 2.1 Library

'References selected (Tools---> References) AND DAO must be higher on the list than ADO.

'***This code was created by viewing YouTubeVideo http://www.youtube.com/watch?v=qHLQTRhU7BE by  ***

'***   599CD  (AccessLearningZone.com)                                                           ***

Dim DB As Database   'set up a database 'container'

Dim rs As Recordset  'set up a recordset 'container'

Set DB = CurrentDb   'identify which database you are putting in the database container

Set rs = DB.OpenRecordset("t_GiftSelections") 'identify which recordset you are putting in the

                                              'recordset container.  This can be a table, a query or

                                              'an SQL statement.

                                              'We are going to add data to the table t_GiftSelections.

rs.AddNew                                     'Identfy action: Add a new record to the recordset.

rs!AdopteeIDf = AdopteeID                     'Tell it that you are making the AdopteeIDf field in the

                                              'recordset/table equal to the AdopteeID value passed

                                              'from previous function

rs!GiftIDf = Me.cboSelectToy                   'Tell it that you are making the GiftIDf field in the

                                              'recordset/table equal to the giftID value on the form

rs.Update                                     'Update the recordset/table

rs.Close                                       'Close the recordset (so no corruption occurs?)

Set rs = Nothing                               'Set the recordset container to nothing (to make sure

                                               'that if it doesn't properly close, then it has nothing

                                               'in it??)

Set DB = Nothing                               'Set the database container to nothing

End Sub
JULIESDB Module
Option Compare Database
Option Explicit
Public Function ConcatenateGifts(AdopteeID As Long) As String

'Make sure this is a public function!!!!

'****************************************************************

'*** This code was based on code Tom Wickerath created for me ***

'****************************************************************

On Error GoTo ProcError

'Purpose:   Return a string containing all the gifts for a specific child.

Dim DB As DAO.Database

Dim rs As DAO.Recordset

Dim strSQL As String

Dim strOut As String

Dim lngLen As Long

Const conSEP = ", "

strSQL = "SELECT GiftName FROM q_GiftLists " _

       & "WHERE [AdopteeIDf] = " & AdopteeID & ";"

Set DB = DBEngine(0)(0)

Set rs = DB.OpenRecordset(strSQL)

    With rs

        Do While Not .EOF

             strOut = strOut & !GiftName & conSEP

            .MoveNext

        Loop

    End With

    rs.Close

    lngLen = Len(strOut) - Len(conSEP)

    If lngLen > 0 Then

        ConcatenateGifts = Left$(strOut, lngLen)

    End If

ExitProc:

   'Cleanup

   On Error Resume Next

   rs.Close: Set rs = Nothing

   DB.Close: Set DB = Nothing

   Exit Function

ProcError:

   MsgBox "Error " & Err.Number & ": " & Err.Description, _

          vbCritical, "Error in ConcatenateGifts Function..."

   Resume ExitProc

End Function

Name

Abby
Doll, Girls Bike

Christine
Doll, Girls Bike

Doug
Boys Bike

Jan
Doll, Car, Girls Bike

Joe
Car, Boys Bike

John
Car, Boys Bike, Horse, Playset

Rose
Car, Girls Bike, Playset

Sam
Boys Bike, Horse
